
PharmaSUG 2017 – Paper AD09 
 

Important & Valuable Things You Can Do with SAS® Metadata 

DICTIONARY Tables and SASHELP Views 
 

Kirk Paul Lafler, Software Intelligence Corporation, Spring Valley, California 
 
 
 

Abstract 

SAS® users can easily and quickly access metadata content with a number of read-only SAS data sets called DICTIONARY tables 

or their counterparts, SASHELP views. During a SAS session, information (known as metadata) is captured including SAS system 

options along with their default values, assigned librefs, table names, column names and attributes, formats, indexes, and 

more. This presentation explores how metadata can be used as input into a SAS code generator or a SAS macro to produce the 

desired results, the application of specific DICTIONARY table and SASHELP view content, examples related to the creation of 

dynamic code, and the creation of a data dictionary. 

 

Introduction 

The SAS System collects and populates valuable information (“metadata”) about SAS libraries, data sets (tables), catalogs, 

indexes, macros, system options, titles, views and a collection of other read-only tables called dictionary tables. Dictionary 

tables serve a special purpose by providing system-related information about the current SAS session’s SAS databases and 

applications.  When a query is requested against a Dictionary table, SAS automatically launches a discovery process at runtime 

to collect information pertinent to that table. This information is made available any time after a SAS session is started. 

 

Dictionary tables and SASHELP views contents permit a SAS session’s activities to be easily accessed and monitored. This 

becomes particularly useful in the design and construction of software applications since the information can be queried and 

the results acted upon in a specific task such as in the allocation of filerefs or librefs. 

 

Tables Used in Examples 

The data used in all the examples in this paper consists of a selection of movies that I’ve viewed over the years, along with 

actors. The Movies table consists of six columns: title, length, category, year, studio, and rating. Title, category, studio, and 

rating are defined as character columns with length and year being defined as numeric columns. The data stored in the Movies 

table is illustrated below. 

 

MOVIES Table 

The data stored in the ACTORS table is illustrated below. 

 

ACTORS Table 

 

 
 

 

 

Exploring SAS Metadata DICTIONARY Tables and SASHELP Views 

SAS users can quickly and conveniently obtain useful information about their SAS session with a number of read-only SAS 

system tables called DICTIONARY tables. At any time during a SAS session, DICTIONARY tables can be accessed using the libref 

DICTIONARY in the FROM clause of a PROC SQL SELECT statement to capture information related to currently defined libnames, 

table names, column names and attributes, formats, and much more. SASHELP views can be accessed using any of your favorite 

procedures or in the DATA step. 

 

SAS 9.1 software supported 22 Dictionary tables and SASHELP views, SAS 9.2 supported 29 Dictionary tables and SASHELP 

views, SAS 9.3 supported 30 DICTIONARY tables and SASHELP views, and SAS 9.4 supports 32 DICTIONARY tables and SASHELP 

views as is illustrated below. 

 

DICTIONARY Tables and SASHELP Views 

DICTIONARY Table SASHELP View Purpose 

CATALOGS VCATALG SAS Catalogs and Catalog-specific Information. 

CHECK_CONSTRAINTS VCHKCON Check Constraints information. 

COLUMNS VCOLUMN Columns from All Tables. 

CONSTRAINT_COLUMN_USAGE VCNCOLU Constraint Column Usage. 

CONSTRAINT_TABLE_USAGE VCNTABU Constraint Table Usage. 

DATAITEMS VDATAIT Information Map Data Items. 

DESTINATIONS VDEST Open ODS Destinations. 

DICTIONARIES VDCTNRY DICTIONARY Tables and their Columns. 

ENGINES VENGINE Available Engines. 

EXTFILES VEXTFL 
Implicitly-defined File Definitions and Files Defined in FILENAME 
statements. 

FILTERS VFILTER Information Map Filters. 

FORMATS VFORMAT Available SAS and User-defined Formats and Informats. 

FUNCTIONS VFUNC Available Functions. 

GOPTIONS VGOPT SAS/GRAPH Software Graphics Options. 

INDEXES VINDEX Information related to Defined Indexes. 

INFOMAPS VINFOMP Information Maps. 

LIBNAMES VLIBNAM Information related to SAS Data Libraries. 

LOCALES VLOCALE Available Locales, Regions, Languages and Currency Symbols. 

MACROS VMACRO Information about Defined Macros. 

MEMBERS VMEMBER Information about SAS Defined Tables, Catalogs and Views. 

OPTIONS VOPTION Information about SAS Default System Options. 

PROMPTS VPROMPT Information about Information Map Prompts. 

PROMPTSXML VPRMXML Information Map Prompts XML. 

REFERENTIAL_CONSTRAINTS VREFCON Information about Referential Constraints. 

REMEMBER VREMEMB All Remembered Information. 

STYLES VSTYLE Information about All Styles. 

TABLES VTABLE SAS Tables and Table-specific Information. 

TABLE_CONSTRAINTS VTABCON Information about Table Constraints. 

TITLES VTITLE Information about Defined Titles. 

VIEWS VVIEW Views and View-specific Information. 

VIEW_SOURCES VSVIEW Sources Referenced by View. 

XATTRS VXATTR Extended Attributes. 

 

 

Displaying DICTIONARY Table Definitions 

A dictionary table’s definition can be displayed by specifying a DESCRIBE TABLE statement. The results of the statements and 

clauses used to create each dictionary table can be displayed on the SAS Log. For example, a DESCRIBE TABLE statement is 

illustrated below to display the CREATE TABLE statement used in building the OPTIONS dictionary table containing current SAS 

System option settings. 

 

PROC SQL Code 

 

PROC SQL; 

  DESCRIBE TABLE 

    DICTIONARY.OPTIONS; 

QUIT; 

SAS Log Results 

 

create table DICTIONARY.OPTIONS 

  ( 

   optname char(32) label='Option Name', 

   setting char(1024) label='Option Setting', 

   optdesc char(160) label='Option Description', 

   level char(8) label='Option Location' 

  ); 

 

Note: The information contained in dictionary tables is also available to DATA and PROC steps outside the SQL procedure. 

Referred to as SASHELP views, each view is prefaced with the letter “V” and may be shortened with abbreviated names. 

SASHELP views can be accessed by referencing the view by its name in the SASHELP library. Please refer to the SAS Procedures 

Guide for further details on accessing and using dictionary views in the SASHELP library. 

 

 

The DICTIONARIES Table and VDCTNRY SASHELP View 

SAS users can identify any new Dictionary table release by accessing the read-only DICTIONARIES Dictionary table or VDCTNRY 

SASHELP view. The contents of the DICTIONARIES Dictionary table and VDCTNRY SASHELP view reveals the names of supported 

tables and views. The following PROC SQL query uses the UNIQUE keyword to generate a listing of existing Dictionary tables. 

 

PROC SQL Code: 

 

PROC SQL ; 

  SELECT UNIQUE MEMNAME 

    FROM DICTIONARY.DICTIONARIES ; 

QUIT ; 

<  or  > 

PROC SQL ; 

  SELECT DISTINCT MEMNAME 

    FROM DICTIONARY.DICTIONARIES ; 

  QUIT ; 

 

 

Dictionary.COLUMNS 

Retrieving information about the columns in one or more data sets or tables is easy with the COLUMNS dictionary table. Similar 

to the results of the CONTENTS procedure, users are able to capture column-level information including column name, type, 

length, position, label, format, informat, and indexes, as well as produce cross-reference listings containing the location of 

columns in a SAS library. For example, the following code requests a cross-reference listing of the tables containing the TITLE 

column in the WORK library.  Note: Care should be used when specifying multiple functions on the WHERE clause since the SQL 

Optimizer is unable to optimize the query resulting in all allocated SAS session librefs being searched. This can cause the query 

to run much longer than expected. 

 

PROC SQL Code 

 

PROC SQL; 

  SELECT * 

    FROM DICTIONARY.COLUMNS 

      WHERE UPCASE(LIBNAME)=”WORK” AND 

            UPCASE(NAME)=”TITLE”; 

QUIT; 

Results 

 

 
 

 

Dictionary.TABLES 

When users need more information about SAS files consider using the TABLES dictionary table. The TABLES dictionary table 

provides detailed information about the library name, member name and type, date created and last modified, number of 

observations, observation length, number of variables, password protection, compression, encryption, number of pages, reuse 

space, buffer size, number of deleted observations, type of indexes, and requirements vector. For example, to obtain a detailed 

list of files in the WORK library, a PROC SQL SELECT query can be constructed as follows. 

 

Note: Because the TABLE Dictionary table produces a considerable amount of information, users should consider specifying a 

WHERE clause when accessing this table. 

 

PROC SQL Code 

 

PROC SQL; 

  SELECT * 

    FROM DICTIONARY.TABLES 

      WHERE UPCASE(LIBNAME)=”WORK”; 

QUIT; 

 

Results 

Creation of a “Custom” Data Dictionary 

SAS users can access the contents of the TABLES (or VTABLE) and COLUMNS (or VCOLUMN) Dictionary tables to create a 

“custom” data dictionary. The following code collects summary-level information on the numeric variables in the MOVIES data 

set; joins the contents of the TABLES and COLUMNS Dictionary tables with the STATS table; and then builds a “custom” report 

using PROC REPORT. 

 

Data Dictionary Listing Code: 

 

proc sql ; 

 create table movies_stats as 

  select "MYDATA" as libname, 

         "MOVIES" as memname, 

         min(year) as Min_Year, 

         max(year) as Max_Year, 

         min(length) as Min_Length, 

         max(length) as Max_Length 

   from mydata.movies ; 

 

 create table joined_tables_columns as 

  select * 

   from dictionary.tables  t, 

        dictionary.columns c, 

        work.movies_stats  s 

    where t.libname = "MYDATA" AND 

          t.memname = "MOVIES" AND 

          c.libname = "MYDATA" AND 

          c.memname = "MOVIES" AND 

          s.libname = "MYDATA" AND 

          s.memname = "MOVIES" ; 

quit ; 

 

TITLE  Data Dictionary Listing ; 

PROC REPORT DATA=work.joined_tables_columns NOWINDOWS ; 

 COLUMNS libname memname crdate modate nobs 

         num_character num_numeric name type length varnum 

         label informat format min_year max_year 

         min_length max_length ; 

 DEFINE libname       / ORDER   'DBMS Libref'       WIDTH=30 ; 

 DEFINE memname       / ORDER   'Table Name'        WIDTH=30 ; 

 DEFINE crdate        / ORDER   'Creation Date'     WIDTH=30 ; 

 DEFINE modate        / ORDER   'Modification Date' WIDTH=30 ; 

 DEFINE nobs          / ORDER   'Number of Rows'    WIDTH=30 ; 

 DEFINE num_character / ORDER   'Number of Character Columns' WIDTH=30 ; 

 DEFINE num_numeric   / ORDER   'Number of Numeric Columns'   WIDTH=30 ; 

 DEFINE name          / DISPLAY 'Column Name'       WIDTH=30 ; 

 DEFINE type          / DISPLAY 'Column Type'       WIDTH=4 

                        CENTER ; 

 DEFINE length        / DISPLAY 'Column Length'     WIDTH=20 ; 

 DEFINE varnum        / DISPLAY 'Column Order'      WIDTH=3 ; 

 DEFINE min_length    / ORDER   'Minimum Movie Length'        WIDTH=3 ; 

 DEFINE max_length    / ORDER   'Maximum Movie Length'        WIDTH=3 ; 

 DEFINE min_year      / ORDER   'Minimum Year'      WIDTH=3 ; 

 DEFINE max_year      / ORDER   'Maximum Year'      WIDTH=3 ; 

RUN ; 

Results 

 

 
 

 
 

 

Conclusion 

The SAS System read-only Dictionary tables and corresponding SASHELP views provide valuable information about SAS libraries, 

data sets, columns and attributes, catalogs, indexes, macros, system options, titles, views, and much more. Users are 

encouraged to research these powerful resources of information to better understand information about data, for the creation 

of system documentation and performance tuning, as well as other important application areas. 

